

# Lección 9

Marcos Bujosa

10 de noviembre de 2023

## Índice

<b>1. Estimación restringida vía mínimos cuadrados restringidos y vía sustitución</b>	<b>2</b>
1.1. Código completo de la práctica <code>GujaratiEx8-3.inp</code> . . . . .	3
<b>2. Test de Chow de cambio estructural</b>	<b>4</b>
2.1. Código completo de la práctica <code>GujaratiSec8-8.inp</code> . . . . .	5

# 1. Estimación restringida vía mínimos cuadrados restringidos y vía sustitución

Guión: [GujaratiEx8-3.inp](#)

Cargue los datos `Table_8.8.gdt` de la pestaña del libro de Gujarati.

`open Table_8.8.gdt`

**Estimación restringida** Suponga que deseamos estimar el siguiente modelo en logaritmos proveniente de una función de Cobb-Douglas:

$$\ln Y = \beta_1 \mathbf{1} + \beta_2 \ln K + \beta_3 \ln L + U;$$

pero que deseamos imponer la restricción de rendimientos constantes a escala, es decir,  $\beta_2 + \beta_3 = 1$ . Veamos dos maneras equivalentes de proceder:

1. Transforme las variables en logaritmos

```
logs GDP Employ Capital
```

- Estime por MCO el modelo sin restringir (guarde el el modelo como icono con el nombre `U` (unrestricted)).

```
# modelo 1
```

```
U <- ols l_GDP const l_Employ l_Capital
```

- Imponga la restricción  $\beta_2 + \beta_3 = 1$ . Desde la ventana del modelo estimado sin restricciones siga los pasos *Contrastes ->Restricciones lineales* y teclee `b[2]+b[3]=1` o bien, en un guión o la consola teclee

```
restrict
```

```
  b[2]+b[3]=1
```

```
end restrict
```

Observe los coeficientes estimados resultantes tras imponer la restricción.

2. Defina las variables `Capital/Employ` y `GDP/Employ` y transforme las nuevas variables mediante logaritmos.

```
series CE = Capital/Employ
```

```
series GDPE = GDP/Employ
```

```
logs CE GDPE
```

- Estime por MCO el modelo

$$\ln Y = \beta_1 \mathbf{1} + \beta_2 \ln \frac{K}{L} + U$$

y compare los resultados anteriores (los del primer modelo tras imponer la restricción).

```
# modelo 2
```

```
R <- ols l_GDPE const l_CE
```

**Contraste de la hipótesis** Calcule el estadístico  $\mathcal{F}$  (en su formulación mediante sumas residuales de los modelos restringidos y sin restringir) y su  $p$ -valor para contrastar la hipótesis de rendimientos constantes a escala. ¿Rechaza la  $H_0$  al 5 % de significación?

```
/* contraste de la F mediante sumas residuales */
```

```
scalar r = 1 # num. restricciones (solo una en este caso)
```

```
scalar f = U.$df*(R.$ess-U.$ess)/U.$ess/r
```

```
pvalue F r U.$df f
```

## 1.1. Código completo de la práctica GujaratiEx8-3.inp

```
open Table_8.8.gdt

logs GDP Employ Capital

# modelo 1
U <- ols l_GDP const l_Employ l_Capital

restrict
  b[2]+b[3]=1
end restrict

series CE = Capital/Employ
series GDPE = GDP/Employ
logs CE GDPE

# modelo 2
R <- ols l_GDPE const l_CE

/* contraste de la F mediante sumas residuales */
scalar r = 1 # num. restricciones (solo una en este caso)
scalar f = U.$df*(R.$ess-U.$ess)/U.$ess/r
pvalue F r U.$df f
```

## 2. Test de Chow de cambio estructural

Guión: GujaratiSec8-8.inp

Cargue los datos Table\_8.9.gdt de la pestaña del libro de Gujarati con datos para la economía americana del 1970 a 1995.

open Table\_8.9.gdt

Considere el modelo:

$$Y = \beta_1 \mathbf{1} + \beta_2 X + U,$$

donde  $X$  es el ahorro de las familias e  $Y$  es la renta disponible.

En el año 1982 se produjo una importante crisis económica. Contraste si el modelo es idéntico para toda la muestra, o si se produjo un cambio estructural (use los periodos 1970–1981 y 1982–1995).

- Estime el modelo restringido (mismos betas para todo el periodo). Guarde la Suma de los Residuos al Cuadrado (SRC)

```
ModeloR      <- ols SAVINGS 0 INCOME --quiet
scalar sse_R = $ess
```

- Estime dos modelos, uno para los 12 primeros datos y otro para los 14 siguientes. Guarde la Suma de los Residuos al Cuadrado (SRC) conjunta del modelo sin restringir.

```
smpl 1 12
ModeloU1     <- ols SAVINGS 0 INCOME --quiet
scalar sse_U1 = $ess
```

```
smpl full
smpl 13 26
ModeloU2     <- ols SAVINGS 0 INCOME --quiet
scalar sse_U2 = $ess
```

- Calcule el estadístico del contraste de cambio estructural de Chow y su p-valor.

```
scalar sse_U = sse_U1+sse_U2
smpl full
scalar chowtest = ((sse_R-sse_U)/2)/((sse_U)/($nobs-4))
pvalue F 2 $nobs-4 chowtest
```

- ¿Rechaza que el modelo es el mismo para todo el periodo? ¿o no?

## 2.1. Código completo de la práctica GujaratiSec8-8.inp

```
open Table_8.9.gdt

ModeloR      <- ols SAVINGS 0 INCOME --quiet
scalar sse_R = $ess

smpl 1 12
ModeloU1     <- ols SAVINGS 0 INCOME --quiet
scalar sse_U1 = $ess

smpl full
smpl 13 26
ModeloU2     <- ols SAVINGS 0 INCOME --quiet
scalar sse_U2 = $ess

scalar sse_U = sse_U1+sse_U2
smpl full
scalar chowtest = ((sse_R-sse_U)/2)/((sse_U/($nobs-4))
pvalue F 2 $nobs-4 chowtest
```